
Abstract

The diagonalization of general mass matrices is a more delicate problem when
eigenvalue degeneracies exist. In this case, often overlooked in the literature, some
difficulties arise related to the freedom in the choice of basis in degenerate subspaces.
Here two simple algorithms are developed to deal with quark and neutrino mass ma-
trices with arbitrary degeneracies.

1 Introduction

In a field theory with fermionic fields ψi, the mass sector of the Lagrangian can be written
as

− Lmass = ψ̄iMijψj + h.c. (1)

The structure of the mass matrix M is constrained by the symmetries of the Lagrangian:
charge conservation, baryon and lepton number conservation, etc. We will consider the
quark and neutrino sectors of the Standard Model with N generations of quarks; nL

left-handed and nR right-handed neutrinos. The mass terms in the quark sector are

− Lmass = ū0
LiM

u
iju

0
Rj + d̄ 0

LiM
d
ijd

0
Rj + h.c., (2)

where u0
L,R (d 0

L,R) are the left and right-handed components of the up (down) quark fields,
and i, j = 1, . . . , N label the different generations of up (down) quarks. TheN -dimensional
mass matrices Mu,d are complex and arbitrary. In practice it is convenient to work in a
quark basis in which Mu,d are diagonal. We make an appropriate change of weak quark
basis

uLi = Uu
Liju

0
Lj , uRi = Uu

Riju
0
Rj , dLi = Ud

Lijd
0
Lj , dRi = Ud

Rijd
0
Rj , (3)

such that the matrices Du ≡ Uu
LM

uUu†
R , Dd ≡ Ud

LM
dUd†

R are diagonal, and rewrite the
mass term as

− Lmass = ūLiD
u
ijuRj + d̄LiD

d
ijdRj + h.c. (4)

Of course this change of basis may affect other terms in the Lagrangian, but we will not
discuss this here. Thus what we denote here as ‘diagonalization’ is really a biunitary
transformation

U1MU †
2 = D (5)

such that D is diagonal (we can also make the matrix elements of D real and positive).

The matrices U1,2 can be calculated [1] noting that (5) and U2M
†U †

1 = D imply

U1MM †U †
1 = D2 , U2M

†MU †
2 = D2 (6)

However, this does not determine U1,2 uniquely if D has degenerate diagonal elements,
and this arbitrariness may cause that D is no longer diagonal: There are matrices U1,2

satisfying (6) that do not necessarily satisfy (5). For instance, let us consider

M =

(

0 1
1 0

)

(7)

If we try to calculate U1,2 by using Eqs. (6), we find MM † = M †M = I, so U1 = U2 = I

and U1MU †
2 is not diagonal. This simple example shows us that we need great care to

deal with the case of degenerate eigenvalues.
The mass terms in the neutrino sector are [2]

− Lmass =
1

2
ψ̄LMνψR + h.c. (8)

with

ψL =

(

νL

(νc)L

)

, ψR =

(

(νc)R

νR

)

, (9)

and Mν a complex (nL + nR) × (nL + nR) symmetric matrix. Under a change of basis

NL = UψL , NR = U∗ψR (10)

the mass term can be written as

− Lmass =
1

2
N̄LDνNR + h.c. (11)

with Dν = UMνU
T a diagonal matrix with real positive elements (it will be shown below

that any complex symmetric matrix Mν can be ‘diagonalized’ with a congruent transfor-
mation of this type. A proof for the nondegenerate case can be found in Ref. [3]). If D is
nondegenerate we can find the matrix U by diagonalizing MνM

†
ν , but if some degeneracies

are present we can get incorrect results.

In this paper we obtain two algorithms for the diagonalization of complex arbitrary
and complex symmetric matrices, respectively, and we write a package Diagon for their use
with Mathematica [4].1 This package makes use of Eigensystem to calculate eigenvalues
and eigenvectors of hermitian and unitary matrices.

2 Algorithms

We will first write an algorithm to diagonalize a complex arbitrary matrix by a biunitary
transformation and then use it to obtain a second one that performs the diagonalization
of a complex symmetric matrix by a congruent transformation. Note that we could use
the first algorithm to diagonalize a complex symmetric matrix with two unitary matrices,
but in general the transformation would not be congruent.

Let M be a complex arbitrary matrix. It is a well-known fact that we can find two
unitary matrices U1, U2 such that U1MU †

2 = D is a diagonal matrix with real positive
elements. We will assume for simplicity that D has diagonal elements di with multiplicity
mi and ordered from lower to higher values. Let V1, V2 be two matrices that diagonalize
MM † and M †M respectively, V1MM †V †

1 = D2, V2M
†MV †

2 = D2. It can be proven that
Vi = KiUi, where Ki are block-diagonal unitary matrices that commute with D

D =









d0Im0

d1Im1

. . .

dkImk









⇒ Ki =










K
(0)
i

K
(1)
i

. . .

K
(k)
i










(12)

(the dimension of the blocks K
(j)
i is mj). Then,

V1MV †
2 = K1DK

†
2 ≡ DB (13)

The matrix in the right-hand side of Eq. (13) is a block-diagonal matrix. Each block is

an unitary matrix times the corresponding eigenvalue of D: D
(j)
B = djK

(j)
1 K

(j)†
2 . Further-

more, from DBD
†
B = D2 we can compute the eigenvalues di and defining λi = 1/di (if

d0 = 0, we take λ0 = 1 and D
(0)
B = Im0

),

Λ = diag(

m0

︷ ︸︸ ︷

λ0, . . . , λ0, . . . ,

mk
︷ ︸︸ ︷

λk, . . . , λk) (14)

1Although Mathematica has a built-in function SingularValues that performs the diagonalization of

an arbitrary matrix by a biunitary transformation, it neglects all zero eigenvalues so it is useless for our

purpose.

we find that U1 ≡ V1, U2 ≡ ΛDBV2 fulfill U1MU †
2 = D.

The diagonalization of a complex symmetric matrix Ms by a congruent transformation
is somewhat more involved. Let U1, U2 be two unitary matrices satisfying U1MsU

†
2 = D,

with D diagonal with real positive elements (we can calculate U1,2 as it is shown above).
As Ms is symmetric, U∗

2MsU
T
1 = D and these equations imply U2 = KU∗

1 , with K a
block-diagonal unitary matrix that commutes with D (analogous to Ki in Eq. (12)).
Then,

U1MsU
T
1 = DK (15)

We now want to decompose K into two unitary symmetric matrices which commute with
D. The simplest way to show it is to see that K is a symmetric matrix (all the blocks
K(j) are symmetric except possibly the block corresponding to d0 = 0, but in this case we
can redefine U2 without changing U1 and take K(0) = I). A complex unitary symmetric
matrix K can always be written as K = eiS , with S a real symmetric matrix [5], and
[K,D] = 0 implies [S,D] = 0, so we can rewrite Eq. (15) as

e−i S

2 U1MsU
T
1 e

−i S

2 = D (16)

However, we need a method to separate the matrix K into two symmetric pieces. First we
diagonalize K with a real orthogonal matrix O: K = OTKdO. It can always be done2 but
we must check that the subroutine we use to calculate eigenvectors gives real results if it is
possible (the built-in functions Eigenvectors and Eigensystem do it). We conventionally

define the square root of K as K1/2 = OTK
1/2
d O, with the eigenvectors in the matrix O

conveniently ordered so that K1/2 commutes with D. Finally, U = (K1/2)†U1 fulfills
UMsU

T = D.

3 Examples

Before we present two examples to illustrate the use of the package Diagon, it is impor-
tant to notice that when the matrix m has degenerate eigenvalues, the set of eigenvectors
calculated with u=Eigenvectors[m] is not necessarily an unitary matrix, and the prod-
uct u.Transpose[Conjugate[u]] may have large off-diagonal elements ǫ ∼ 10−1. This is
not a roundoff error (typically of order O(10−16)) as we show in the example below with
an exact diagonalization. We use DiagonalizeH, a subroutine to diagonalize hermitian
matrices that uses Eigensystem and the Gram-Schmidt method of orthonormalization in
each degenerate subspace to yield orthonormal eigenvectors. The following example will
illustrate this. The numerical results have been obtained running Mathematica 2.2.2 for
Linux.

2Note that S is a real symmetric matrix

We write a previously calculated hermitian matrix m with eigenvalues (0, 0, 1, 1) and
diagonalize it. Note that DiagonalizeH does not return the eigenvectors but their complex
conjugate.

In[1]:= <<Diagon.m

In[2]:= m=1/4 {{1,(1-I)/Sqrt[2],(-1+I)/Sqrt[2],1},{(1+I)/Sqrt[2],2,

-1+I,(-1+I)/Sqrt[2]},{(-1-I)/Sqrt[2],-1-I,2,(-1+I)/Sqrt[2]},

{1,(-1-I)/Sqrt[2],(-1-I)/Sqrt[2],3}};

In[3]:= u=Chop[Eigenvectors[m]];

In[4]:= u.Adj[u]

Out[4]= {{8, (-2 + 2 I) Sqrt[2], 0, 0}, {(-2 - 2 I) Sqrt[2], 4, 0, 0},

8 2 2 I 2 2 I 12

> {0, 0, -, (- - ---) Sqrt[2]}, {0, 0, (- + ---) Sqrt[2], --}}

5 5 5 5 5 5

In[5]:= Simplify[Conjugate[u].m.Transpose[u]]

8 2 2 I

Out[5]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, -, (- + ---) Sqrt[2]},

5 5 5

2 2 I 12

> {0, 0, (- - ---) Sqrt[2], --}}

5 5 5

In[6]:= u2=DiagonalizeH[m][[2]];

In[7]:= Chop[u2.Adj[u2]]

Out[7]= {{1., 0, 0, 0}, {0, 1., 0, 0}, {0, 0, 1., 0}, {0, 0, 0, 1.}}

In[8]:= Chop[u2.N[m].Adj[u2]]

Out[8]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 1., 0}, {0, 0, 0, 1.}}

The package Diagon (see the Appendix for a complete description) contains the subroutines
DiagonalizeH, Diagonalize and DiagonalizeS . It has been tested more than 100 times
on Mathematica 2.2.2 for Linux and Mathematica 2.2 for Windows. We have used the
instructions listed below with different sets of eigenvalues, obtaining always correct results.
First we load the package and define the matrices m and ms to test Diagonalize and
DiagonalizeS respectively. We use 16-digit precision in the input matrices.

In[9]:= SeedRandom[]

In[10]:= m1=Table[Random[Complex,{-1-I,1+I}],{i,8},{j,8}];

In[11]:= m2=Table[Random[Complex,{-1-I,1+I}],{i,8},{j,8}];

In[12]:= ma1=m1+Adj[m1];

In[13]:= ma2=m2+Adj[m2];

In[14]:= mu1=Chop[Eigenvectors[ma1]];

In[15]:= mu2=Chop[Eigenvectors[ma2]];

In[16]:= m=mu1.DiagonalMatrix[{0,0,1,1,1,1,2,2}].mu2;

In[17]:= ms=mu1.DiagonalMatrix[{0,0,1,1,1,1,2,2}].Transpose[mu1];

In[18]:= sol=Diagonalize[m];

In[19]:= sol[[1]]

Out[19]= {0, 0, 1., 1., 1., 1., 2., 2.}

In[20]:= Chop[sol[[2]].m.Adj[sol[[3]]]]

Out[20]= {{0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0},

> {0, 0, 1., 0, 0, 0, 0, 0}, {0, 0, 0, 1., 0, 0, 0, 0},

> {0, 0, 0, 0, 1., 0, 0, 0}, {0, 0, 0, 0, 0, 1., 0, 0},

> {0, 0, 0, 0, 0, 0, 2., 0}, {0, 0, 0, 0, 0, 0, 0, 2.}}

In[21]:= sol2=DiagonalizeS[ms];

In[22]:= sol2[[1]]

Out[22]= {0, 0, 1., 1., 1., 1., 2., 2.}

In[23]:= Chop[sol2[[2]].ms.Transpose[sol2[[2]]]]

Out[23]= {{0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0},

> {0, 0, 1., 0, 0, 0, 0, 0}, {0, 0, 0, 1., 0, 0, 0, 0},

> {0, 0, 0, 0, 1., 0, 0, 0}, {0, 0, 0, 0, 0, 1., 0, 0},

> {0, 0, 0, 0, 0, 0, 2., 0}, {0, 0, 0, 0, 0, 0, 0, 2.}}

A Appendix

(* Title: Diagon *)

(* Author: J. A. Aguilar-Saavedra *)

(* Summary:

This package provides some algorithms for the diagonalization

of complex arbitrary and complex symmetric matrices

*)

(* Version: 1.1 *)

(*

This package is available by anonymous ftp at deneb.ugr.es

in directory pub/packages/

*)

BeginPackage["Diagon‘","LinearAlgebra‘Orthogonalization‘"]

Adj::usage =

"Adj[m] gives Transpose[Conjugate[m]]"

Ortho::usage =

"Ortho[{v,u}] takes a list of eigenvalues v and eigenvectors u and

returns a list {v,u’} consisting of the eigenvalues v and

orthonormal eigenvectors u’"

DiagonalizeH::usage =

"DiagonalizeH[m] gives a list {v,u}, with the eigenvalues and

eigenvectors of the square matrix m, such that

u.m.Transpose[Conjugate[u]]=DiagonalMatrix[v]. The list of

eigenvalues v is ordered by increasing value"

Diagonalize::usage =

"Diagonalize[m] gives a list {v,u1,u2} where v is a row vector, and

u1,u2 are matrices fulfilling and u1.m.Transpose[Conjugate[u2]]=

DiagonalMatrix[v]"

DiagonalizeS::usage =

"DiagonalizeS[m] gives a list {v,u} where v is a row vector and u is

a matrix fulfilling u.m.Transpose[u]=DiagonalMatrix[v]"

Begin["‘Private‘"]

(* Definitions for internal use only *)

(* Increase the dimension of a matrix with the identity *)

IncDim1[m_]:=Transpose[Prepend[Transpose[Prepend[m,Table[0,

{Length[m]}]]],Join[{1},Table[0,{Length[m]}]]]]

IncDim[m_,nn_]:=Nest[IncDim1,m,nn]

(* Pick up a submatrix *)

DecDim[m_,nn_]:=Drop[Transpose[Drop[Transpose[m],nn]],nn]

(* Main definitions *)

Adj[m_]:=Transpose[Conjugate[m]];

(* Note that the scalar product that works is linear in the

first variable and antilinear in the second:

InnerProduct->(Dot[Conjugate[#2],#1]&) *)

Ortho[sys_]:=Block[{i,j,allvalues,vectors,values,orthvectors},(

allvalues=N[Chop[sys[[1]]]];

vectors=N[Chop[sys[[2]]]];

For[i=1,i<=Length[allvalues],i++,

For[j=i+1,j<=Length[allvalues],j++,

If[Chop[allvalues[[i]]-allvalues[[j]]]==0,

allvalues[[j]]=allvalues[[i]] ] ] ];

values=Union[allvalues];

orthvectors=Flatten[Table[GramSchmidt[vectors[[Flatten[

Position[allvalues,values[[i]]]]]],

InnerProduct->(Dot[Conjugate[#2],#1]&)],{i,Length[values]}],1];

{Sort[allvalues],Chop[orthvectors]}

)];

DiagonalizeH[m_]:=Conjugate[

Ortho[Eigensystem[N[(m+Adj[m])/2] ] ] ]/;MatrixQ[m] &&

Length[m]==Length[Transpose[m]]

Diagonalize[m_]:=Block[{i,sol1,sol2,v,nz,mbox0,mbox,mod,mod2},(

sol1=DiagonalizeH[m.Adj[m]];

sol2=DiagonalizeH[Adj[m].m];

v=Chop[Sqrt[sol1[[1]]]];

nz=Count[v,0];

mbox0=Chop[sol1[[2]].m.Adj[sol2[[2]]]];

mbox=DecDim[mbox0,nz];

mod=Chop[mbox.Transpose[Conjugate[mbox]]];

mod2=DiagonalMatrix[Table[1/Abs[Sqrt[mod[[i,i]]]],{i,Length[mod]}]];

{v,Chop[sol1[[2]]],Chop[IncDim[mbox.mod2,nz].sol2[[2]]]}

)]/;MatrixQ[m] && Length[m]==Length[Transpose[m]] &&

Complement[Flatten[Chop[m]],{0}]!={}

Diagonalize[m_]:={Table[0,{Length[m]}],IdentityMatrix[Length[m]],

IdentityMatrix[Length[m]]}/;MatrixQ[m] &&

Length[m]==Length[Transpose[m]] &&

Complement[Flatten[Chop[m]],{0}]=={}

DiagonalizeS[m_]:=Block[{sol,v,u1,u2,nz,k0,k,sol2,rot,kd},(

sol=Diagonalize[m];

v=Chop[sol[[1]]];

u1=sol[[2]];

u2=sol[[3]];

nz=Count[v,0];

k0=Chop[Conjugate[u2].Adj[u1]];

k=DecDim[k0,nz];

sol2=Ortho[Eigensystem[k]];

rot=Sort[Conjugate[sol2[[2]]],OrderedQ[{Abs[#2],Abs[#1]}]& ];

kd=Chop[rot.k.Adj[rot]];

{v,Chop[IncDim[Adj[rot].Sqrt[kd].rot,nz].u1]}

)]/;MatrixQ[m] && Length[m]==Length[Transpose[m]] &&

Complement[Flatten[Chop[m]],{0}]!={}

DiagonalizeS[m_]:={Table[0,{Length[m]}],

IdentityMatrix[Length[m]]}/;MatrixQ[m] &&

Length[m]==Length[Transpose[m]] &&

Complement[Flatten[Chop[m]],{0}]=={}

End[]

EndPackage[]

